# PRÁCTICA 3:
# Integrales Triples

Para calcular integrales triples haremos uso del teorema de Fubini.

$$a \le x \le b, \quad \phi_1(x) \le y \le \phi_2(x) \quad \text{y} \quad \gamma_1(x,y) \le z \le \gamma_2(x,y) \qquad (1)$$

$$c \le y \le d, \quad \psi_1(y) \le x \le \psi_2(y) \quad \text{y} \quad \gamma_1(x,y) \le z \le \gamma_2(x,y) \qquad (2)$$

**Teorema de Fubini en regiones de integración de $\mathbb{R}^3$**

*Si suponemos que $\Omega$ es del tipo (I), entonces*

$$\iiint_\Omega f(x,y,z)dxdydz = \int_a^b \left( \int_{\phi_1(x)}^{\phi_2(x)} \left( \int_{\gamma_1(x,y)}^{\gamma_2(x,y)} f(x,y,z)dz \right) dy \right) dx$$

*o*

$$\iiint_\Omega f(x,y,z)dxdydz = \int_c^d \left( \int_{\psi_1(y)}^{\psi_2(y)} \left( \int_{\gamma_1(x,y)}^{\gamma_2(x,y)} f(x,y,z)dz \right) dx \right) dy$$

*siempre que $\Omega$ esté definida por la ecuación (1) o por la ecuación (2).*

- Para calcular las integrales iteradas utilizaremos la orden **integrate.**
- Cuando no podamos hallar la integral de forma exacta utilizaremos la orden **quad_qags**, para calcularla de forma aproximada.

Como vimos en la práctica anterior, el resultado que obtenemos con esta orden es una lista con cuatro elementos: el valor aproximado de la integral, la estimación del error, el número de veces que se tuvo que evaluar el integrando y finalmente un código de error que será cero si no surgieron problemas.

- Para indicar si una expresión es positiva o negativa utilizaremos la orden **assume.**
- Para representar gráficamente las superficies, dadas en forma implícita, cargaremos previamente el paquete correspondiente mediante la orden **load(draw)** y utilizaremos la orden **wxdraw3d.**
- Vimos que en dicha orden tenemos las opciones siguientes:**enhanced3d=true**: colorea las superficies,**surface_hide=true**:no dibuja la parte no visible de las superficies.

**Ejemplo 1**

*Calcular $\iiint_B z^2 y e^x \, dxdydz$ donde $B$ es el paralelepípedo definido por las relaciones*

$$0 \le x \le 1, \quad 1 \le y \le 2, \quad -1 \le z \le 1$$

**Solución con wxMaxima:**

```
(%i1)   iterada1:integrate(z^2*y*%e^x,x,0,1);
```
(iterada1) $(\%e-1)\,y\,z^2$

```
(%i2)   iterada2:integrate(iterada1, y, 1, 2);
```
(iterada2) $\frac{3\,(\%e-1)\,z^2}{2}$

```
(%i3)   integrate(iterada2,z,-1,1);
```
(%o3) $\%e-1$

**Ejemplo 2**

Calcular la integral triple $\iiint_\Omega (x^2+y^2)dxdydz$ siendo $\Omega$ la región acotada por el paraboloide $z = x^2 + y^2$ y el plano $z = 2$.

**Solución con wxMaxima:**

```
(%i5)   load(draw)$
0 errors, 0 warnings
```

```
(%i6)   wxdraw3d(enhanced3d=true,surface_hide=true,
        implicit(x^2+y^2=z,x,-2,2,y,-2,2,z,0,2));
```

(%t6)




(%o6)

```
(%i7) wxdraw2d(proportional_axes=xy,
          implicit(x^2+y^2=2,x,-2,2,y,-2,2));
```

(%t7)




(%o7)

```
(%i8) iterada1:integrate( x^2+y^2,z,x^2+y^2,2);
```

(iterada1) $\left(-y^2-x^2+2\right)\left(y^2+x^2\right)$

```
(%i9) iterada2:integrate(iterada1,
      y,-sqrt(2-x^2),sqrt(2-x^2));
```

(iterada2) $-\dfrac{2\sqrt{2-x^2}\left(8x^4-12x^2-8\right)}{15}$

```
(%i10) integrate(iterada2, x, -sqrt(2), sqrt(2));
```

(%o10) $\dfrac{4\pi}{3}$

En muchas ocasiones para poder calcular las integrales triples tendremos que utilizar cambios de variables.

- Para calcular la matriz Jacobiana utilizaremos la orden **jacobian** y para el determinante la orden **determinant.**

**Teorema del cambio de variables en integrales triples**

*Dada la integral triple* $\iiint_{\Omega} f\,(x, y, z)\,dxdydz$ *el problema consiste en* cambiar las variables $x$, $y$ ,$z$ de la función $f(x, y, z)$ por tres nuevas variables, digamos $u$, $v$,$w$ según las ecuaciones de transformación

$$x = x(u, v, w), \quad y = y(u, v, w), z = z(u, v, w)$$

$$\iiint_{\Omega} f(x,y,z)dxdydz = \iiint_{\Omega'} f(\overline{F}(u,v,w))\,|J(u,v,w)|\,dudvdw$$

$$J(u,v,w) = \frac{\partial(x,y,z)}{\partial(u,v,w)} = \det\begin{bmatrix} \frac{\partial x}{\partial u} & \frac{\partial x}{\partial v} & \frac{\partial x}{\partial w} \\ \frac{\partial y}{\partial u} & \frac{\partial y}{\partial v} & \frac{\partial y}{\partial w} \\ \frac{\partial z}{\partial u} & \frac{\partial z}{\partial v} & \frac{\partial z}{\partial w} \end{bmatrix}$$

$$\frac{\partial(x,y,z)}{\partial(u,v,w)} = \frac{1}{\frac{\partial(u,v,w)}{\partial(x,y,z)}}$$

**Ejemplo 3**

Calcular el volumen del paralelepípedo limitado por los 6 planos en $\mathbb{R}^3$: $x+y+z=1$; $x+y+z=2$; $x-y-z=1$; $x-y-z=2$; $x+y-z=1$; $x+y-z=2$.
Si introducimos las nuevas variables $u, v, w$ tales que

$$u = x+y+z, v = x-y-z, w = x+y-z$$

la región $\Omega'$ del espacio $uvw$ que corresponde a nuestra región $\Omega$ se describe de una manera muy sencilla. De hecho se tiene

$$\Omega' = \{(u,v,w)\,/1 \leq u \leq 2, 1 \leq v \leq 2, 1 \leq w \leq 2\}$$

**Solución con wxMaxima**:

```
(%i1)   kill(all)$

-->     /* Ecuaciones del cambio de variable */;

(%i3)   u(x,y,z):=x+y+z;
        v(x,y,z):=x-y-z;
        w(x,y,z):=x+y-z;
(%o1)   u(x,y,z):=x+y+z
(%o2)   v(x,y,z):=x-y-z
(%o3)   w(x,y,z):=x+y-z

-->     /* Cálculo de la matriz jacobiana */

(%i4)   m:jacobian ([u(x,y,z), v(x,y,z),w(x,y,z)], [x, y,z]);
0 errors, 0 warnings
```

$$(m)\quad \begin{bmatrix} 1 & 1 & 1 \\ 1 & -1 & -1 \\ 1 & 1 & -1 \end{bmatrix}$$

```
-->     /* Cálculo del jacobiano */
```

```
(%i5)   j:1/(abs(determinant(m)));
(j)     1/4

-->     /* Cálculo de la integral triple */

(%i6)   iterada1:integrate(j,u,1,2);
(iterada1)  1/4
(%i7)   iterada2:integrate(iterada1,v,1,2);
(iterada2)  1/4
(%i8)   integrate(iterada2,w,1,2);
(%o8)   1/4
```

Los dos cambios de variables más utilizados son el cambio a coordenadas cilíndricas y el cambio a coordenadas esféricas.

**Coordenadas cilíndricas**

Vamos a introducir tres nuevas coordenadas para un punto $P = (x, y, z)$ en $\mathbb{R}^3$, denotadas por $r, \theta, z$, según las fórmulas

$$x = r\cos\theta, \quad y = r\,\mathrm{sen}\,\theta, \quad z = z$$




- Para simplificar el jaconiano y las integrales iteradas utilizaremos la orden **trigsimp.**

**Ejemplo 4**

Calcular la integral triple $\iiint_\Omega x^2\,dxdydz$ siendo $\Omega$ la región acotada por el cono $z = \sqrt{x^2 + y^2}$ y el plano $z = 1$.

**Solución con wxMaxima:**

```
(%i1)   kill(all)$
```

(%i1)
```
wxdraw3d(enhanced3d=true,surface_hide=true,
implicit(sqrt(x^2+y^2)=z,x,-2,2,y,-2,2,z,0,1));
```
0 errors, 0 warnings

(%t1) 


(%o1)

(%i2)
```
wxdraw2d(proportional_axes=xy,
    implicit(x^2+y^2=1,x,-1,1,y,-1,1));
```

(%t2) 


(%o2)

```
-->     /* Ecuaciones del cambio de variable */;

(%i5)   x(r,t,z):=r*cos(t);
        y(r,t,z):=r*sin(t);
        z(r,t,z):=z;
```

(%o3) $x(r,t,z):=r\cos(t)$

(%o4) $y(r,t,z):=r\sin(t)$

(%o5) $z(r,t,z):=z$

```
-->     /* Cálculo de la matriz jacobiana */;

(%i6)   m:jacobian([x(r,t,z),y(r,t,z),z(r,t,z)],[r,t,z]);
```

(m) $\begin{bmatrix} \cos(t) & -r\sin(t) & 0 \\ \sin(t) & r\cos(t) & 0 \\ 0 & 0 & 1 \end{bmatrix}$

```
-->     /* Cálculo del jacobiano */

(%i7)   j:determinant(m);
```

(j) $r\sin(t)^2+r\cos(t)^2$

```
(%i8)   j:trigsimp(j);
```

(j) $r$

```
-->     /* Cálculo de la integral triple */

(%i9)   iterada1:integrate((x(r,t,z))^2*j,z,r,1);
```

(iterada1) $(1-r)\,r^3\cos(t)^2$

```
(%i10)  iterada2:integrate(iterada1,r,0,1);
```

(iterada2) $\frac{\cos(t)^2}{20}$

```
(%i11)  integrate(iterada2,t,0,2*%pi);
```

(%o11) $\frac{\pi}{20}$

**Coordenadas esféricas**
Vamos a introducir ahora tres nuevas coordenadas para un punto $P = (x, y, z)$ en $\mathbb{R}^3$, denotadas por $\rho, \theta, \phi$, según las fórmulas

$$x = \rho \cos\theta \operatorname{sen}\phi, \quad y = \rho \operatorname{sen}\theta \operatorname{sen}\phi, \quad z = \rho\cos\phi$$




## Ejemplo 5

Calcular la integral triple $\iiint_\Omega (x^2+y^2+z^2)\,dxdydz$ siendo $\Omega$ la región limitada por la esfera $x^2+y^2+z^2=2$ y el cono $z=\sqrt{x^2+y^2}$ .
a) Utilizando coordenadas cartesianas.
b) Utilizando coordenadas cilíndricas.
c) Utilizando coordenadas esféricas.

Eliminando $z$ de las ecuaciones de las superficies:

$$\begin{matrix} x^2+y^2+z^2=2 \\ z=\sqrt{x^2+y^2} \end{matrix} \quad \rightarrow \left\{ \begin{matrix} x^2+y^2+z^2=2 \\ z^2=x^2+y^2 \end{matrix} \right. \quad \rightarrow 2x^2+2y^2=2$$

vemos que la proyección sobre el plano $xy$ está limitada por la circunferencia

$$x^2+y^2=1$$

Por tanto las variaciones de las variables cartesianas son:

$$\sqrt{x^2+y^2} \le z \le \sqrt{2-x^2-y^2} \; ; -\sqrt{1-x^2} \le y \le \sqrt{1-x^2} \; ; \; -1 \le x \le 1$$

En cilíndricas son:

$$r \le z \le \sqrt{2-r^2} \; ; 0 \le r \le 1 \; ; \; 0 \le \theta \le 2\pi$$

En esféricas son:

$$0 \le \rho \le \sqrt{2} \; ; \; 0 \le \phi \le \frac{\pi}{4} \; ; \; 0 \le \theta \le 2\pi$$

**Solución con wxMaxima:**

```
(%i1)    kill(all)$
```

```
(%i1) wxdraw3d(
      implicit(sqrt(x^2+y^2)=z,x,-2,2,y,-2,2,z,0,2),
      implicit(x^2+y^2+z^2=2,x,-2,2,y,-2,2,z,0,2));
0 errors, 0 warnings
```

(%t1)




(%o1)

```
(%i2) wxdraw2d(proportional_axes=xy,
        implicit(x^2+y^2=1,x,-1,1,y,-1,1));
```

(%t2)




(%o2)

```
-->   /* Solución en cartesianas */;
(%i3) assume(2-x^2-y^2>0);
```

(%o3) $[-y^2-x^2+2>0]$

(%i5)
```
iterada1:integrate( x^2+y^2+z^2,z,
    sqrt(x^2+y^2),sqrt(2-x^2-y^2));
```

(iterada1) $$\frac{\sqrt{-y^2-x^2+2}\left(2\,y^2+2\,x^2+2\right)}{3}-\frac{\sqrt{y^2+x^2}\left(4\,y^2+4\,x^2\right)}{3}$$

(%i6)
```
assume(1-x^2>0);
```

(%o6) $[x^2<1]$

(%i7)
```
iterada2:integrate(iterada1,
y,-sqrt(1-x^2),sqrt(1-x^2));
```

(iterada2) $$-\frac{2\,x^4\,\mathrm{asinh}\left(\frac{\sqrt{1-x^2}}{|x|}\right)+\left(4-x^4\right)\mathrm{asin}\left(\frac{\sqrt{1-x^2}\,\sqrt{2-x^2}}{x^2-2}\right)+x^2\sqrt{1-x^2}}{2}$$

(%i8)
```
2*quad_qags(iterada2, x, 0, 1);
```

(%o8) $[2.082064455302322, 7.822920958842029\,10^{-9}, 462, 0]$

-->
```
/* Solución en cilíndricas */;
```

(%i11)
```
x(r,t,z):=r*cos(t);
y(r,t,z):=r*sin(t);
z(r,t,z):=z;
```

(%o9) $x(r,t,z):=r\cos(t)$

(%o10) $y(r,t,z):=r\sin(t)$

(%o11) $z(r,t,z):=z$

(%i12)
```
m:jacobian([x(r,t,z),y(r,t,z),z(r,t,z)],[r,t,z]);
```

(m) $$\begin{bmatrix}\cos(t) & -r\sin(t) & 0\\ \sin(t) & r\cos(t) & 0\\ 0 & 0 & 1\end{bmatrix}$$

(%i13)
```
j:determinant(m);
```

(j) $r\sin(t)^2+r\cos(t)^2$

(%i14)
```
j:trigsimp(j);
```

(j) $r$

(%i15)
```
iterada1:integrate(((x(r,t,z))^2+(y(r,t,z))^2+(z(r,t,z))^2)*j,
z,r,sqrt(2-r^2));
```

(iterada1) $$r\left(\frac{\sqrt{2-r^2}\left(3\,r^2\sin(t)^2+3\,r^2\cos(t)^2-r^2+2\right)}{3}-\frac{3\,r^3\sin(t)^2+3\,r^3\cos(t)^2+r^3}{3}\right)$$

(%i19)
```
iterada2:integrate(iterada1,r,0,1);
```

(iterada2) $$\frac{2^{7/2}\sin(t)^2+2^{7/2}\cos(t)^2+2^{5/2}}{15}-\frac{10\sin(t)^2+10\cos(t)^2+2}{15}$$

```
(%i20) iterada2:trigsimp(iterada2);
```

(iterada2) $$\frac{2^{5/2}-4}{5}$$

```
(%i21) integrate(iterada2,t,0,2*%pi);
```

(%o21) $$\frac{2\left(2^{5/2}-4\right)\pi}{5}$$

```
(%i22) %,numer;
```

(%o22) 2.0820644553097

```
--> /* Solución en esféricas */;
```

```
(%i25) x(r,t,fi):=r*cos(t)*sin(fi);
       y(r,t,fi):=r*sin(t)*sin(fi);
       z(r,t,fi):=r*cos(fi);
```

(%o23) $$\mathrm{x}(r,t,fi):=r\cos(t)\sin(fi)$$

(%o24) $$\mathrm{y}(r,t,fi):=r\sin(t)\sin(fi)$$

(%o25) $$\mathrm{z}(r,t,fi):=r\cos(fi)$$

```
(%i26) m:jacobian([x(r,t,fi),y(r,t,fi),z(r,t,fi)],[r,fi,t]);
```

(m) $$\begin{bmatrix} \sin(fi)\cos(t) & \cos(fi)\,r\cos(t) & -\sin(fi)\,r\sin(t) \\ \sin(fi)\sin(t) & \cos(fi)\,r\sin(t) & \sin(fi)\,r\cos(t) \\ \cos(fi) & -\sin(fi)\,r & 0 \end{bmatrix}$$

```
(%i27) j:determinant(m);
```

(j) $$-\sin(fi)\,r\sin(t)\left(-\sin(fi)^2\,r\sin(t)-\cos(fi)^2\,r\sin(t)\right)+\sin(fi)^3\,r^2\cos(t)^2+\cos(fi)^2\sin(fi)\,r^2\cos(t)^2$$

```
(%i28) j:trigsimp(j);
```

(j) $$\sin(fi)\,r^2$$

```
(%i29) iterada1:integrate(((x(r,t,fi))^2+(y(r,t,fi))^2+(z(r,t,fi))^2)*j,
       r,0,sqrt(2));
```

(iterada1) $$\frac{\sin(fi)\left(2^{5/2}\sin(fi)^2\sin(t)^2+2^{5/2}\sin(fi)^2\cos(t)^2+2^{5/2}\cos(fi)^2\right)}{5}$$

```
(%i30) iterada2:integrate(iterada1,fi,0,%pi/4);
```

(iterada2) $$\frac{\frac{2^{7/2}\sin(t)^2+2^{7/2}\cos(t)^2+2^{5/2}}{3}-\frac{10\sin(t)^2+10\cos(t)^2+2}{3}}{5}$$

```
(%i31) iterada2:trigsimp(iterada2);
```

(iterada2) $$\frac{2^{5/2}-4}{5}$$

```
(%i35) ratprint:false$
(%i37) integrate(iterada2,t,0,2*%pi),numer;
```

(%o37) 2.0820644553097

## EJERCICIOS PROPUESTOS

**1.**

Calcular la integral triple $\iiint_\Omega xyz\,dxdydz$ siendo $\Omega$ la región limitada por la esfera $x^2+y^2+z^2=9$, contenida en el primer octante.

a) Utilizando coordenadas cartesianas.
b) Utilizando coordenadas cilíndricas.
c) Utilizando coordenadas esféricas.

**Solución:** $\frac{243}{16}$

**2.**

Calcular el volumen del cuerpo limitado por los paraboloides

$$z=x^2+y^2 \quad ; \quad z=2-x^2-y^2$$

a) Utilizando coordenadas cartesianas.
b) Utilizando coordenadas cilíndricas.

**Solución:** $\pi$

**3.**

Hallar el centro de gravedad del cuerpo homogéneo $\Omega$ limitado por el paraboloide $z=x^2+y^2$ y el plano $z=1$.

**Solución:** $xg=0, yg=0, zg=\frac{2}{3}$

**4.**

Hallar el momento de inercia $I_z$ del sólido $\Omega$, situado por encima del plano $xy$, limitado por el paraboloide $z=x^2+y^2$ y el cilindro $x^2+y^2=1$, suponiendo que la densidad de masa es $\mu(x,y,z)=x^2+y^2$.

**Solución:** $\frac{\pi}{4}$

**5.**

Calcular la integral triple $\iiint_\Omega \sqrt{1-x^2-y^2-z^2}\,dxdydz$ siendo $\Omega$ la región limitada por la esfera $x^2+y^2+z^2=1$.

**Solución:** $\frac{\pi^2}{4}$

**6.**

Calcular la integral triple $\iiint_\Omega (x^2+y^3+y^2z^2)dxdydz$ siendo $\Omega$ el sólido limitado por las superficies: $z=1-x^2-y^2$; $x^2+y^2-x=0$; $z=0$.a) En coordenadas cartesianas. b) En coordenadas cilíndricas y comprobar el resultado.
Nota: Representar gráficamente las superficies.

**Solución:** $\frac{87\pi}{2560}$

**7.**

Calcular la integral triple $\iiint_\Omega (x^2y^2 + y^6z^6)dxdydz$ siendo $\Omega$ el sólido limitado por las superficies: $z = 1 - \sqrt{x^2+y^2}$; $x^2 + y^2 - x = 0$; $z = 0$.
a) En coordenadas cartesianas. b) En coordenada cilíndricas, hallando previamente el jacobiano. c) Comprobar que el resultado es el mismo.
Nota: Representar gráficamente las superficies.

**Solución: 0.0037631575866479**

**8.**

Calcular la integral triple $\iiint_\Omega x^2y^2z^2dxdydz$ siendo $\Omega$ el sólido limitado por las superficies: $z = 2 - \sqrt{x^2+y^2}$; $x^2 + y^2 - 2y = 0$; $z = 0$.
a) En coordenadas cartesianas. b) En coordenada cilíndricas, hallando previamente el jacobiano. c) Comprobar que el resultado es el mismo.
Nota: Representar gráficamente las superficies.

**Solución: 0.085180403610931**